# $Id: PKGBUILD 102633 2010-12-09 08:40:59Z andrea $
# Maintainer: Antonio Rojas <arojas@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: Pierre Schmitz <pierre@archlinux.de>

pkgbase=polkit-qt
pkgname=(polkit-qt5)
pkgver=0.112.0+git20180107
pkgrel=2
pkgdesc='A library that allows developers to access PolicyKit API with a nice Qt-style API'
arch=(x86_64)
url='https://www.kde.org/'
license=(LGPL)
depends=(polkit qt5-base)
makedepends=(cmake git)
#source=("http://download.kde.org/stable/apps/KDE4.x/admin/${pkgbase}-1-${pkgver}.tar.bz2")
source=("git+https://anongit.kde.org/polkit-qt-1.git#commit=cbecf60ac7b5c17cf37503d6675bf5542efbab29")
sha256sums=('SKIP')

prepare() {
  mkdir -p build
}

build() {
  cd build
  cmake ../$pkgbase-1 \
    -DCMAKE_INSTALL_PREFIX=/usr \
    -DLIB_DESTINATION=/usr/lib
  make
}

package_polkit-qt5() {
  cd build
  make DESTDIR="$pkgdir" install
}
